

Lecture Notes – Session 14 – IS201 - Exam Review

Brandt Redd – 2020-03-02

Area

```
Sub Area()  
    Dim height As Double  
    Dim width As Double  
  
    'Clear the output in case of error  
    Range("B3").ClearContents  
  
    'Check for missing height  
    If IsEmpty(Range("B1").Value) Then  
        MsgBox "Height required!"  
        Exit Sub  
    End If  
  
    'Check for missing width  
    If IsEmpty(Range("B2").Value) Then  
        MsgBox "Width required!"  
        Exit Sub  
    End If  
  
    'Read the input  
    height = Range("B1").Value  
    width = Range("B2").Value  
  
    'Generate the output  
    Range("B3").Value = height * width  
End Sub
```

*These are the solutions to the problems in
ExamReview.xlsx available at
<https://brandtredd.org/files/IS201/ExamReview.xlsx>*

Sentence

```
Sub Sentence()  
    Dim name As String  
    Dim temp As Double  
    Dim attire As String  
  
    'Get the inputs  
    name = Range("B2").Value  
    temp = Range("B3").Value  
  
    'Determine the attire  
    If temp <= Range("D2").Value Then  
        attire = Range("E2").Value  
    ElseIf temp <= Range("D3").Value Then  
        attire = Range("E3").Value  
    ElseIf temp <= Range("D4").Value Then  
        attire = Range("E4").Value  
    ElseIf temp <= Range("D5").Value Then  
        attire = Range("E5").Value  
    Else  
        attire = Range("E6").Value  
    End If  
  
    'Compose the sentence ( Can also use WorksheetFunction.Proper() )  
    Range("B8").Value = StrConv(name, vbProperCase) & " wore a " & LCase(attire) & _  
        " because it was " & Trim(Str(temp)) & " degrees outside."  
End Sub
```

Sentence Using Loop

```
Sub Sentence()  
    Dim name As String  
    Dim temp As Double  
    Dim attire As String  
    Dim i As Integer  
  
    'Get the inputs  
    name = Range("B2").Value  
    temp = Range("B3").Value  
  
    'Determine the attire  
    For i = 2 To 6  
        If temp <= Cells(i, 4) Then  
            Exit For  
        End If  
    Next i  
    attire = Cells(i, 5).Value  
  
    'Compose the sentence  
    Range("B8").Value = StrConv(name, vbProperCase) & " wore a " & LCase(attire) & "  
because it was " & Trim(Str(temp)) & " degrees outside."  
    ' Could also use WorksheetFunction.Proper()  
End Sub
```

Vehicles

```
Sub Translate()  
    'Variables for label translation  
    Dim lastRow As Integer  
    Dim letter As String  
    Dim i As Integer, j As Integer  
    Dim label As String  
  
    'Variables for length calculations  
    Dim length As Integer  
    Dim count As Integer  
    Dim sum As Long  
    Dim min As Integer  
    Dim max As Integer  
  
    'Clear the counts  
    For i = 2 To 6  
        Cells(i, 5).Value = 0  
    Next i  
  
    'Find the end of the input  
    lastRow = Cells(2, 1).End(xlDown).Row  
  
    'Clear summary info  
    count = 0  
    sum = 0  
    min = 1000  
    max = 0  
  
    For i = 2 To lastRow  
        'Convert letter to label  
        letter = LCase(Left(Cells(i, 1), 1))  
        j = 2  
        Do Until i = 6 Or letter = LCase(Left(Cells(j, 4), 1))  
            j = j + 1  
        Loop  
        label = Cells(j, 4).Value  
        Cells(i, 2) = label  
  
        'Increment the count  
        Cells(j, 5) = Cells(j, 5) + 1  
  
        'Accumulate the summary info  
        length = Len(label)  
        count = count + 1  
        sum = sum + length  
        If min > length Then  
            min = length  
        End If  
        If max < length Then  
            max = length  
        End If  
    Next i  
  
    'Report Summary  
    Range("H2").Value = min  
    Range("H3").Value = max  
    Range("H4").Value = sum / count  
  
End Sub
```